Copy No. 101

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION

NASA PROGRAM APOLLO WORKING PAPER NO. 1317

BREMSSTRAHLUNG DOSE CALCULATIONS FOR APOLLO MANNED SPACECRAFT

[redacted]

MANNED SPACECRAFT CENTER

HOUSTON, TEXAS

June 21, 1967

NASA PROGRAM APOLLO WORKING PAPER NO. 1317

BREMSSTRAHLUNG DOSE CALCULATIONS FOR APOLLO MANNED SPACECRAFT

PREPARED BY

Joseph W. Snyder, Charles Sterling Portwood, III,
Donald A. Witt, Timothy T. White, and Alva C. Hardy

AUTHORIZED FOR DISTRIBUTION

W N Hess

Wilmot N. Hess
Director of Science and Applications

NATIONAL AERONAUTICS AND SPACE ADMINISTRATION
MANNED SPACECRAFT CENTER
HOUSTON, TEXAS
June 21, 1967

PRECEDING PAGE BLANK NOT FILMED.

CONTENTS

FIGURES

SYMBOLS

A	area
a,c,h	distances in figure 2
D	dose
E	electron kinetic energy
e^-	electron
F	electron flux
f	fraction of bremsstrahlung entering the shield
I	energy of bremsstrahlung produced
J	electron current
j	electron directional flux
K	proportionality constant
k	0.0004/MeV
p	radius of shield
R	electron range
r	distance along electron path
S	bremsstrahlung energy flux
T	thickness of shield
t	time
x	distance along normal to shield
Z	absorber atomic number
α	angle gamma ray path makes with the electron path
β	electron velocity divided by the speed of light
γ	gamma ray

μ	shield attenuation coefficient
Ω	solid angle
ψ	azimuthal angle around electron path
ψ_m	angle at which the gamma ray is emitted parallel to the shield surface
σ	tissue attenuation coefficient
θ	angle electron path makes with the normal to the shield

ABSTRACT

Astronauts and equipment orbiting the earth within the Van Allen Radiation Belts are subjected to bremsstrahlung radiation; this radiation is produced when electrons are decelerated by interaction with matter. The purpose of this paper is to develop a capability to evaluate the bremsstrahlung hazard to Apollo astronauts; computer programs have been written for predicting this component of the total dose. These theoretical calculations will be compared in a later paper with bremsstrahlung measurements obtained from GT-10.

BREMSSTRAHLUNG DOSE CALCULATIONS

FOR APOLLO MANNED SPACECRAFT

By Joseph W. Snyder, Charles Sterling Portwood, III, Donald A. Witt, Timothy T. White, and Alva C. Hardy

SUMMARY

Bremsstrahlung, X-rays covering a continuous spectrum, is produced during deceleration of electrons as they collide with atomic nuclei. Astronauts encounter bremsstrahlung in the Van Allen Radiation Belts, where electrons are decelerated in spacecraft shielding; the subsequent bremsstrahlung component of total dose is important for mission planning and crew safety. The purpose of this study is to develop an ability to calculate the bremsstrahlung doses within the Apollo Command and Service Modules (CSM) and Lunar Module (LM) during future missions. Specifically, the effort has been directed toward the Apollo program, but the techniques are applicable to all missions within the Van Allen Belts. The computer program, SPW20, was developed to generate dose versus thickness curves for monoenergetic electrons. The output of SPW20 and that of the North American Aviation Geometry Code are used in conjunction with MSC-Radiation and Fields Branch dose calculational programs and the MSC Orbital Dose Code to compute specific mission doses.

THEORY

Introduction

Bremsstrahlung is energy lost in the form of X-rays by charged particles as a result of inelastic collisions with atomic nuclei. The energy loss is significant for very light particles such as electrons. The total energy of bremsstrahlung emitted by a particle of initial energy E_o which is completely stopped in an absorber of atomic number Z is given by

$$I_o = kZE_o^2 \tag{1}$$

where k is taken to be 0.0004/MeV (ref. 1).

The electron range is determined primarily by ionization energy losses since bremsstrahlung accounts for only a very small fraction of the electron energy. The range of an electron is approximately proportional to its energy; thus, the electron slows down nearly uniformly and most of the bremsstrahlung is produced near the surface where the electron enters.

First Approximation

The following assumptions are made to simplify calculations:

1. The electron flux F (electrons/cm^2/sec) is omnidirectional, isotropic, and monoenergetic with energy E_o (MeV).

2. The shield is a spherical shell, outer radius p, thickness T, with a dose point at the center.

3. An electron hitting the shell gives off I_o (MeV) in bremsstrahlung at the surface.

4. The bremsstrahlung is created isotropically.

The first objective is to calculate the energy flux S (MeV/cm^2/sec) at the dose point, which can easily be converted to dose. The current of electrons hitting the shell is defined as

$$J_s = \int j(\Omega) A_{eff}(\Omega) d\Omega \tag{2}$$

where j is the flux in a given direction, A_{eff} is the cross-sectional area in that direction, and Ω is solid angle. Under these assumptions, this becomes

$$J_s = \int \left(\frac{F}{4\pi}\right)\left(\pi p^2\right) d\Omega = \pi p^2 F (\text{elec/sec}) \tag{3}$$

Since I_o is the energy from each electron, the energy flux S' of the spherical X-ray source is just

$$S' = \frac{J_s I_o}{4\pi p^2} = \frac{1}{4} F I_o \left(\text{MeV/cm}^2\text{/sec}\right) \tag{4}$$

If the average attenuation coefficient over all bremsstrahlung energies for the shield is μ, then

$$S = S'e^{-\mu T} = \frac{1}{4} FI_o e^{-\mu T} \tag{5}$$

Depth Production

The first modification to the simple model above was to account for bremsstrahlung being produced at various depths in the shield, hence being attenuated less. Since in a spacecraft configuration $p >> T$ and $p >> R$, the electron range, any small section of the shield may be treated as a plane slab. The following notation will be used (see fig. 1): θ = angle of incoming electron with the normal to the shield, r = distance along the electron path, x = distance along normal to shield, $x = r = 0$ at the surface. It will be assumed that $T \geq R$ (although this only changes the limit on some integrations) and that the electron travels in a straight line stopping at $r = R$. The next objective will be to calculate the profile dI/dx which describes the bremsstrahlung produced per unit depth.



Figure 1.- Depth production geometry.

By assuming the electron range proportional to its energy, the energy of an electron is

$$E = E_o(1 - r/R) \tag{6}$$

and the bremsstrahlung it will give off is

$$I = kZE^2 \tag{7}$$

Then, since $r = x \sec \theta$ (see fig. 1)

$$\frac{dI}{dx}(\theta) = \left(\frac{dI}{dE}\right)\left(-\frac{dE}{dr}\right)\left(\frac{dr}{dx}\right) = (2kZE)\left(\frac{E_o}{R}\right)(\sec \theta)$$

$$\frac{dI}{dx}(\theta) = \frac{2I_o}{R}(1 - r/R)(\sec\theta) \tag{8}$$

Now consider some small area A (cm^2) on the surface. The current of electrons hitting A is

$$J_a = \int j(\Omega) A \cos\theta \, d\Omega = \frac{AF}{4\pi}\int_o^{2\pi} d\phi \int_o^{\pi/2} \sin\theta \, d(\sin\theta) = \frac{AF}{4} \tag{9}$$

Averaging over all particles

$$\frac{dI}{dx} = \frac{1}{J_a}\int \frac{dI}{dx}(\theta) j(\Omega) A_{eff}(\Omega) d \; = 2\int_o^{\pi/2} \cdot \left[\frac{dI}{dx}(\theta)\cos\theta \sin\theta\right] d\theta$$

$$= \frac{4I_o}{R}\int_o^{\cos^{-1}\left(\frac{x}{R}\right)} \left(1 - \frac{x}{R}\sec\theta\right) \sec\theta \cos\theta \sin\theta \, d\theta$$

$$= \frac{4I_o}{R}\int_o^{\cos^{-1}\left(\frac{x}{R}\right)} \left[\sin\theta \, d\theta + \frac{x}{R}\frac{d(\cos\theta)}{\cos\theta}\right]$$

$$\frac{dI}{dx} = \frac{4I_o}{R}\left(1 - \frac{x}{R} + \frac{x}{R}\ln\left|\frac{x}{R}\right|\right) \tag{10}$$

One may readily show that $I_o = \int_o^R \left(\frac{dI}{dx}\right) dx$, as it should. Now, instead of one spherical source, there are many infinitesimal ones having an energy flux

$$dS' = \frac{J_s dI}{4\pi(p - x)^2} = \frac{J_s\left(\frac{dI}{dx}\right)dx}{4\pi p^2} \quad \text{since} \quad p >> T > x \tag{11}$$

Finally, the energy flux at the dose point is

$$S = \int e^{-\mu(T-x)} dS' = \frac{1}{4} F I_o e^{-\mu T}\int_o^R \frac{4}{R} e^{\mu x}\left(1 - \frac{x}{R} + \frac{x}{R}\ln\left|\frac{x}{R}\right|\right) dx \tag{12}$$

which was used in our actual computations.

Anisotropic Correction

The assumption that bremsstrahlung is produced isotropically is not very good for relativistic electrons ($E_0 > 1$ MeV). The angular distribution (ref. 2) is given by

$$\frac{dI}{d\Omega} = \frac{K \sin^2\alpha}{(1 - \beta \cos \alpha)^6}$$

where K is a proportionality constant, α is the angle from the electron direction and β is the electron velocity divided by the speed of light. The first step chosen to correct for this anisotropic distribution was to define a correction factor which is the ratio of the fraction of bremsstrahlung produced in the inward direction ($0 \leq \theta \leq \pi/2$) in the actual case to that in the isotropic case. In the actual case we can write the fraction inward for any given θ as

$$f(\theta) = \frac{\int_{in} \left(\frac{dI}{d\Omega}\right) d\Omega}{\int \left(\frac{dI}{d\Omega}\right) d\Omega} \tag{13}$$

In evaluating this integral, it is convenient to use an α-ψ cylindrical system where $\alpha = 0$ is the electron direction and ψ is the azimuthal angle around that direction. The meanings of x and θ are the same as before. The Z-axis is in the plane defined by the X-axis and the electron direction, with X-axis lying at $\psi = 0$. The Y-Z plane is parallel to the shield surface. The integral can be broken into three parts:

a. If $0 \leq \alpha \leq \pi/2 - \theta$, any angle ψ is pointed inward.

b. If $\pi/2 + \theta \leq \alpha \leq \pi$, any angle ψ is pointed outward.

c. If $\pi/2 - \theta \leq \alpha \leq \pi/2 + \theta$, only the range $-\psi_m \leq \psi \leq \psi_m$ is pointed inward, where ψ_m is the value of ψ where the X-ray (γ) is emitted parallel to the surface (see fig. 2).

From figure 2

$$\sec \psi_m = -\sec\left(\pi-\psi_m\right) = -\frac{c}{a} = -\frac{c}{h}\frac{h}{a} = -\tan \alpha \tan \theta \tag{14}$$

Writing equation (13) according to these parts with part (b) = 0

$$f(\theta) = \frac{\int_0^{2\pi} d\psi \int_0^{(\pi/2)-\theta} \frac{dI}{d\Omega} \sin\alpha\, d\alpha + \int_{-\psi_m}^{\psi_m} d\psi \int_{(\pi/2)-\theta}^{(\pi/2)+\theta} \frac{dI}{d\Omega} \sin\alpha\, d\alpha}{\int_0^{2\pi} d\psi \int_0^{\pi} \frac{dI}{d\Omega} \sin\alpha\, d\alpha}$$

$$f(\theta) = \frac{\int_0^{(\pi/2)-\theta} \frac{dI}{d\Omega} \sin\alpha\, d\alpha + \frac{\psi_m}{\pi} \int_{(\pi/2)-\theta}^{(\pi/2)+\theta} \frac{dI}{d\Omega} \sin\alpha\, d\alpha}{\int_0^{\pi} \frac{dI}{d\Omega} \sin\alpha\, d\alpha} \tag{15}$$



Figure 2.- Anisotropic bremsstrahlung production geometry.

In the isotropic case the inward fraction is obviously 1/2, so the correction factor can then be written as

$$f_c = \frac{\int_0^{\pi/2} f(\theta) d\theta}{\int_0^{\pi/2} \left(\frac{1}{2}\right) d\theta}$$

$$= \frac{4}{\pi} \int_0^{\pi/2} f(\theta) d\theta \tag{16}$$

which was used in the actual computations.

Dose Conversion

Having the energy flux S from equation (12), one may convert to dose rate by multiplying by σ, the cross section of tissue for the bremsstrahlung (averaged over energy), and an appropriate unit conversion factor. With σ in cm^2/g, the conversion factor is 1.6×10^{-8}, so that the dose rate is

$$\frac{dD}{dt} = 1.6 \times 10^{-8} \sigma f_c S (rad/sec) \tag{17}$$

SPACECRAFT SHIELDING BREAKDOWN

The geometry code, developed by North American Aviation, Inc., used by MSC in radiation-dose computations utilizes a ray tracing method to calculate shield thickness as a function of solid angle for any point within a spacecraft or other shielding configuration. The spacecraft geometry is defined with all components represented by hexahedrons, cylinders, spheres, hemispheres, cones, truncated cones, ellipsoids, and truncated ellipsoids. These elemental volumes are defined as being either solid or void shields with each having its material composition and density identified. Complex spacecraft components can be described by combinations of different elemental shapes, with void shields deleting the interior or other portions of the solid shields. Figure 3 illustrates the description of one such component. The computer program calculates thicknesses of the intersected shields along a given ray, converts to equivalent thickness in gm/cm^2 of a predefined standard material, and sums the results to obtain the total shield thickness for the ray or solid angle.

The program considers self-absorption of the crew by the inclusion of up to six phantom men similar to the one shown in figure 4. These phantom men are used optionally, and may be placed in the spacecraft in various standing or sitting positions.

The program was designed primarily for use in computations of the attenuation of high-energy protons and other heavy particles; however, it has been adapted for use with electrons and other types of penetrating radiation.

BREMSSTRAHLUNG DOSE CALCULATIONS

In order to support the manned spaceflight program with information concerning the radiation hazards of a particular spacecraft trajectory, the MSC Orbital Dose Code (MODC) was developed to utilize existing environment and dose-calculational programs in obtaining estimates of the dose received while in earth orbit. The radiation hazard which affects earth-orbital manned spaceflight is due in part to naturally occurring electrons and protons which are trapped by the earth's magnetic field. The problem of trapped radiation was intensified by the Starfish high-altitude nuclear detonation of July 9, 1962, which artificially injected electrons into the trapping magnetic field. The intensity of this artificial radiation is expected to decrease exponentially with time, but is still of significance when assessing radiation hazards. In addition to the radiation from primary electrons and protons, bremsstrahlung radiation is also considered.



Figure 3.- Description of spacecraft component by combinations of elemental shapes.



Figure 4.- Phantom man.

The bremsstrahlung dose for a given spacecraft and trajectory is determined in the MODC by calculating the instantaneous dose rate for numerous points along the trajectory and then integrating numerically with respect to time. The time interval between points must be small enough so that the integral can be adequately computed by the trapezoidal rule. The calculational method used in determining bremsstrahlung dose is given by

$$\text{Bremmsstrahlung dose} = \sum_i F_i Dk_i Dt_i \int_{E_{co}}^{E_{max}} (dN/dE)_i D(E)\, dE$$

where

F_i = the integral electron flux associated with spatial location of the ith point on the trajectory

Dk_i = the fraction the electron flux has decreased due to an exponential decay with time of the artificially injected electrons for the ith point

Dt_i = the time interval between the ith and the ith-1 point

E_{max} = upper limit of integration

E_{co} = the average minimum energy of the incident electrons sufficient to reach the dose point

$(dN/dE)_i$ = the differential of the normalized integral electron energy spectrum for the ith point

$D(E)$ = bremsstrahlung dose as a function of incident electron energy for a given spacecraft geometry

dE = differential electron energy

The electron flux and energy spectra used by the MODC are taken from a recent trapped radiation environment model (ref. 3). The Dk_i's are calculated using the MSC "Decay Program" and the D(E) array was generated by MSC - Radiation and Fields Branch dose-calculational programs,[1] which utilize the results of SPW20.

[1]See section SPACECRAFT SHIELDING BREAKDOWN.

CONCLUSIONS

The computer program SPW20 (see appendix A) produced dose versus thickness curves for monoenergetic electrons. The results of this program are shown in figure 5, normalized to one electron per square centimeter. The incident electron energy is at the right of each curve.

To compare our results with Keller and Pruett (ref. 4), the program BREMDOSE (see appendix B) was written. The results of this program are shown in figure 6 and are within a factor of 1.5 of similar graphs in reference 2. The soft spectra is given by

$$F(E) = 1.724 \exp\left(-\frac{E}{0.58}\right)$$

which is somewhat similar to Van Allen Belt spectra. The fission spectrum is given by

$$F(E) = 0.71 \exp\left(-0.575E - 0.055E^2\right)$$

Because of good agreement between our results and others, using the graphs in figure 5 to estimate the bremsstrahlung dose for Apollo seems justified.



Figure 5.- Bremsstrahlung dose vs thickness from SPW20 for monoenergetic electrons.



Figure 6.- Bremsstrahlung doses for specific electron spectra.

PRECEDING PAGE BLANK NOT FILMED.

APPENDIX A

PRECEDING PAGE BLANK NOT FILMED.


```
      PROGRAM SSW 2
      COMMON MU(700),SIGMA(700)
      REAL K,IO,MU
      DELXK=.01
      K=0.0004
      Z=13.
      DOSECONV=1.6E-8
      F=1.
      CALL SIGMAMU
C     THIS SUBROUTINE CALCULATES MU AND SIGMA
      PRINT 8
    8 FORMAT(* ENERGY(MEV)  THICKNESS(GM/SQCM)  DOSERATE(RADS/SEC/ELEC)*
     1)
      DO 50, J=1,14
      EO=.5*J
      CALL BREMFWRD(FC,EO)
      R=EO/1.65
   10 IO=K*Z*EO*EO
C STATEMENT 10 IS EQUIVALENT TO EQUATION (1)
      PRINT 510,FC,EO,IO
C THE NEXT SECTION CALCULATES SS=SIGMA*S AT A PARTICULAR BREMSSTRAHLUNG
C ENERGY XK AND WEIGHTS IT BY A NORMALIZED SPECTRUM FACTOR FOFXK. THE
C SPECTRUM SHAPE WAS FROM EVANS, THE ATOMIC NUCLEUS, P.615
      T=1
   15 SS=0
      M=1
      XK=.01
   20 FOFXK=2.*(1.-XK/EO)/EO
      CALL GETSI(R,T,MU(M),SI)
C THIS SUBROUTINE CALCULATES SI, THE INTEGRAL IN EQUATION (12)
      SS=SS+F*IO/R*EXP(-MU(M)*T)*SI*FOFXK* DELXK*SIGMA(M)
      XK=XK+DELXK
      M=M+1
      IF (XK.LT.EO)GO TO 20
      DOSERATE=DOSECONV*FC*SS
      PRINT 501,EO,T,DOSERATE
  501 FORMAT(*      *F4.1,*             *F4.1,*        *E25.4)
      T=T+1
      IF(T.LT.1 )GO TO 15
   50 CONTINUE
  510 FORMAT(///// 7(1 F11.2//))
      CALL EXIT
      END
```

```
      SUBROUTINE SIGMAMU
C THIS SUBROUTINE CALCULATES VALUE OF MU AND SIGMA FROM EMPIRICAL EQUATIONS
C FITTED TO CROSS SECTION DATA GIVEN IN EVANS, THE ATOMIC NUCLEUS,PP.714-715
      COMMON MU(700),SIGMA(700)
      REAL LYA,LYP,MU
      YB=1.11E-5 $ YM=-3.158 $ YC=.028 $ LYA=-.346 $ LYP=-2.4
      XM=-3.30
      XB=1.096E-6
      D=3.298E-2
      XLA=-.301
      XLP=-1.83
      DO 777 J=1,700
      XK=.01*J
      YH=(ALOG10(XK)-LYA)**2/(2.*LYP)
      MU(J)=YB*XK**YM+YC*10.**YH
      H=(ALOG10(XK)-XLA)**2/(2*XLP)
  777 SIGMA(J)=XB*XK**XM+D*10.**H
      RETURN
      END
```

```
      SUBROUTINE BREMFWRD(FC,E)
      DIDOMEGA(ALPHA)=SINF(ALPHA)**2/(1.-B*COSF(ALPHA))**6
      FC=0
      B=SQRT(1-(.511/(E+.511))**2)
      PI=3.1416
      DTHETA=DALPHA=.001*PI
      THETA=-.5*DTHETA
      DO 40 J=1,500
      THETA=THETA+DTHETA
      COTTHETA=COTF(THETA)
      ALIM1=.5*PI-THETA
      ALIM2=.5*PI+THETA
      SUM1=SUM2=SUM3=0
      ALPHA=-.5*DALPHA
      DO 30 K=1,1000
      ALPHA=ALPHA+DALPHA
      TERM=DIDOMEGA(ALPHA)*SINF(ALPHA)
      IF(ALPHA-ALIM1)20,20,10
   10 IF(ALPHA.GT.ALIM2)GO TO 30
      ARG=COTF(ALPHA)*COTTHETA
      IF(ARG.LT.-1.)ARG=-1.
      PSIMAX=COSF(-ARG)
      SUM2=SUM2+PSIMAX*TERM/PI
      GO TO 30
   20 SUM1=SUM1+TERM
   30 SUM3=SUM3+TERM
   40 FC=FC+2.*(SUM1+SUM2)/SUM3*DTHETA
C STATEMENT 40 IS EQUIVALENT TO EQUATION (16)
C SUM1,SUM2,AND SUM3 ARE THE THREE INTEGRALS IN EQUATION (15)
      RETURN
      END
```

```
      SUBROUTINE GETSI(R,T,XMU,SI)
C THIS SUBROUTINE CALCULATES SI, THE INTEGRAL IN EQUATION (12).
C WHEN THE RANGE IS GREATER THAN THE SHIELD THICKNESS, THEN T REPLACES R AS
C THE LIMIT OF INTEGRATION
      F(Y)=1-Y+Y*LOGF(Y)
      XLIM=R
      IF(R.GT.T)XLIM=T
      SI=0
      X=.005*XLIM
      DX=2.*X
      DO 10 J=1,100
      SI=SI+F(X/R)*EXP(XMU*X)*DX
   10 X=X+DX
      RETURN
      END
```

APPENDIX B

PRECEDING PAGE BLANK NOT FILMED.


```
      PROGRAM BREMDOSE
      DIMENSION D(14,10),F(50),E(50)
      READ 402,DELE,E0
  402 FORMAT(2F10.2)
      READ 401 ,((D(K,J),J=1,10),K=1,14)
  401 FORMAT(5E10.4)
      I=1
      E(I)=E0+.5*DELE
   10 I=I+1
      E(I)=E(I-1)+DELE
      IF(E(I).GT..7)GO TO 12
      GO TO 10
   12 N=I-1
      L=1
      DO 15 I=1,N
   15 F(I)=1.72*EXP(-E(I)/.58)
   16 DO 50 J=1,10
      EI=J
      DOSE=0
      DO 25 I=1,N
      ET=EI-.5
   20 ET=EI+.5
      IF(E(I).GT.ET)GO TO 20
      K=2.*EI
   25 DOSE=DOSE+F(I)*DELE*2.*((E(I)-ET+.5)*D(K,J)+(ET-E(I))*D(K-1,J))
   50 PRINT 501,J,DOSE
  501 FORMAT(///* NORMALIZED DOSE FOR*I3,* G/CM2 IS * E10.3)
      IF(L.EQ.2)GO TO 80
      DO 60 I=1,N
   60 F(I)=.11*EXP(-E(I)*(.575+.055*E(I)))
      L=2
      GO TO 16
   80 CALL EXIT
      END
```

REFERENCES

1. Baggerly, L. L.; Dance, W. E.; Forman, B. J.; and Johnson, J. H.: Bremsstrahlung Production in Thick Aluminum and Iron Targets by 0.5 to 3.0 MeV Electrons. Second Symposium on Protection Against Radiations in Space (NASA SP-71), Gatlinburg, Tennessee, 1964, p. 452.

2. Flügge, S., ed.: Handbuch der Physik. Berlin, vol. 30, 1957, p. 363.

3. Vette, J. I.; Lucero, A. B.; and Wright, J. A.: Models of the Trapped Radiation Environment. NASA SP-3024.

4. Keller, F. L.; and Pruett, R. G.: The Effect of Charged Particle Environments on Manned Military Space Systems. Second Symposium on Protection Against Radiations in Space (NASA SP-71), Gatlinburg, Tennessee, 1965, p. 280.